

SAS Code

```

/* First Cycle Only Models */
  * logistic regression;
  proc genmod data=firstIVF descending;
    class age_quart(ref='0' param=ref) infdx1 (ref='3' param=ref);
    model livebirth = age_quart bmi infdx1 eversmk/ dist=binomial link=logit;
  run;
  * log binomial regression;
  proc genmod data=firstIVF descending;
    class age_quart(ref='0' param=ref) infdx1 (ref='3' param=ref);
    model livebirth =age_quart bmi infdx1 eversmk/ dist=binomial link=log;
  run;
/* Multiple Cycle Models */
  * Mixed Effects Model: The Modified Poisson Approach;
  proc glimmix data=IVF;
    class idnum age_quart(ref='0') infdx1 (ref='3');
    model livebirth= age_quart bmi infdx1 eversmk/ link=log
    dist=poisson solution cl; *Note: to fit the log binomial
    model, simply specify dist=binomial instead of dist=poisson;
    random intercept / subject=idnum type=un;
    lsmeans age_quart / cl; *the lsmeans statement requests the
    probabilities of live birth for each quartile of age, at the
    average value of all covariates;
  run;
  * Unweighted GEE, Log Binomial;
  proc genmod data=IVF descending;
    class idnum age_quart(ref='0' param=ref) infdx1 (ref='3'
    param=ref);
    model livebirth= age_quart bmi infdx1 eversmk/ dist=binomial
    link=log;
    repeated subject=idnum/ type=un corrw;
  run;
  * GEE Models - Weighted, Log Binomial;
  proc genmod data=IVF descending;
    class idnum age_quart(ref='0' param=ref) infdx1 (ref='3'
    param=ref);
    model livebirth= age_quart bmi infdx1 eversmk/ dist=binomial
    link=log;
    repeated subject=idnum/ type=un;
    weight w;
  run;
/* Obtaining the Working Correlation Matrix */
  * Correlation Matrix for the Unadjusted, Unweighted GEE, Unstructured;
  proc genmod data=IVF descending;
    class idnum age_quart(ref='0' param=ref);
    model livebirth= age_quart/ dist=binomial link=log;
    repeated subject=idnum/ type=un corrw;
  run; *corrw in the repeated statement requests the working
  correlation matrix;
  * Correlation Matrix for the Unadjusted, Unweighted GEE, Compound Symmetry;
  proc genmod data=IVF descending;
    class idnum age_quart(ref='0' param=ref);
    model livebirth= age_quart/ dist=binomial link=log;
    repeated subject=idnum/ type=cs corrw;
  run;

```